



INTERNATIONAL ATOMIC ENERGY AGENCY
UNITED NATIONS EDUCATIONAL, SCIENTIFIC AND CULTURAL ORGANIZATION



INTERNATIONAL CENTRE FOR THEORETICAL PHYSICS

34100 TRIESTE (ITALY) - P.O.B. 688 - MIRAMARE - STRADA COSTIERA 11 - TELEPHONE: 2240-1
CABLE: CENTRATOM - TELEX 400302 - I

SECOND SCHOOL ON ADVANCED TECHNIQUES
IN COMPUTATIONAL PHYSICS
(18 January - 12 February 1988)

SMR.262/15

MODULA - 2 AND PASCAL

PART 1: MODULA - 2

V.B.A.FACK
University of Ghent, Belgium

MODULA-2 REFERENCE CARD

INITIALISATION

Before using MODULA-2: `uselng` to initialise environment.

After using MODULA-2: `uselng remove` to restore environment.

COMPILING AND LINKING

To compile: `m2c name[.def]`

To link: `m2l name`

RUNNING

`m2 name`

COMPILING AND LINKING WITH mod EDITOR

`mod name[.mod]`

`mod name.def`

Editing

<ESC>	exit mod
	escape from display boxes
<F1>	help
<F3>	load file
<F4>	save file

Cursor positioning

<←>	character left
<→>	character right
<↑>	line up
<↓>	line down
<PG DN>	page up
<PG UP>	page down
<CTRL>-<A>	to beginning of line
<CTRL>-<E>	to end of line
<HOME>	to beginning of text
<END>	to end of text

Deleting and inserting text

<F8>	Starts selection of a block of text, at current cursor position. Use cursor positioning commands to mark end of selected text. Selected text is highlighted.
	If block of text is selected, deletes it and puts it in 'scratchpad'. Otherwise, deletes character under cursor.
<INS>	Inserts contents of 'scratchpad' at current cursor position.
<BACKSPACE>	Deletes character before cursor.

Search and replace

<F10>	allows to specify a search string
<CTRL>-<S>	forward search of specified string
<CTRL>-<R>	backward search of specified string
<CTRL>-<Q>	replaces old string with new string

Searches and replaces start at current cursor position.

Windowing

<ALT>-<F10> window menu, with a.o. following commands:

close	close current window
hor. split	split current window horizontally at cursor position
vert.split	split current window vertically at cursor position
full screen	make current window full screen window

Select option by typing first letter of command.

<F7> switch between windows

Syntax checking

Press <F2> to check syntax. Either an error message is displayed at the first error, or a message that no syntax errors were found.

Press <ESC> to continue.

Compiling

Press <F5> to compile.

After compilation with errors, press <ALT>-<F5> to position cursor at next error and display error message. Otherwise press <ESC> to continue.

Linking

Press <F6> to link.

After linking, press <ESC> twice to exit mod and return to MS-DOS.

1

Program reads a sequence of positive integers until -1 is read, computes maximum and minimum of the sequence and output the numbers read and their differences with this maximum and minimum.

```
PROGRAM EXAM1
C Fortran version
PARAMETER (NMAX=100)
DIMENSION NUM(NMAX)
READ *, NUM(1)
NR=2
MAX=NUM(1)
MIN=NUM(1)
20 READ *, NUM(NR)
IF (NUM(NR).EQ.-1) GOTO 10
IF (NUM(NR).GT.MAX) THEN
    MAX=NUM(NR)
ELSE IF (NUM(NR).LT.MIN) THEN
    MIN=NUM(NR)
END IF
NR=NR+1
GOTO 20
10 DO 30 I=1,NR-1
30 PRINT *, NUM(I), MAX-NUM(I), NUM(I)-MIN
STOP
END
```

```

. {$B-,U+,R+}
. PROGRAM Example1 (input, output);
. {Pascal version}
. CONST NMax = 100;
. VAR Nr, i, Max, Min : INTEGER;
.     Number           : ARRAY [1..Nmax] OF INTEGER;
. BEGIN
.   Nr := 1; Max := 0; Min := MAXINT;
.   Read (Number[Nr]);
.   WHILE Number[Nr] <> -1 DO
.     BEGIN
.       IF Number[Nr] > Max THEN
.         Max := Number[Nr]
.       ELSE IF Number[Nr] < Min THEN
.         Min := Number[Nr];
.       Nr := Nr + 1;
.       Read (Number[Nr])
.     END (* while *);
.   FOR i := 1 TO Nr - 1 DO
.     BEGIN
.       Write (Number[i]:6, Max - Number[i]:6,
.             Number[i] - Min:6);
.       WriteLn
.     END (* for *)
.   END.

```

```

. MODULE Example1;
. (* Modula-2 version *)
. FROM InOut IMPORT
.   ReadInt, WriteInt, WriteLn;
. CONST
.   NMax = 100;
. VAR
.   Nr, i, Max, Min : INTEGER;
.   Number           : ARRAY [1..NMax] OF INTEGER;
. BEGIN
.   ReadInt (Number[1]);
.   Nr := 2;
.   Max := Number[1]; Min := Number[1];
.   ReadInt (Number[Nr]);
.   WHILE Number[Nr] <> -1 DO
.     IF Number[Nr] > Max THEN Max := Number[Nr]
.     ELSIF Number[Nr] < Min THEN Min := Number[Nr]
.     END (* if *);
.     INC (Nr);
.     ReadInt (Number[Nr])
.   END (* while *);
.   FOR i := 1 TO Nr - 1 DO
.     WriteInt (Number[i], 6);
.     WriteInt (Max - Number[i], 6);
.     WriteInt (Number[i] - Min, 6); WriteLn
.   END (* for *)
. END Example1.

```

```

MODULE Example2;
(* Count the occurrences of the letters 'a'..'z'
   in a line of text, followed by '.' *)
FROM InOut IMPORT
  Read, Write, WriteCard, WriteString, WriteLn;
TYPE
  Letters      = ['a'..'z'];
  Occurrences = ARRAY Letters OF CARDINAL;
VAR
  Occ : Occurrences;
  Ch  : CHAR;
BEGIN
  FOR Ch := 'a' TO 'z' DO Occ[Ch] := 0 END;
  LOOP
    Read (Ch); Write (Ch);
    IF Ch = '.' THEN
      EXIT
    ELSIF (Ch >= 'a') AND (Ch <= 'z') THEN
      INC (Occ[Ch])
    END (* if *)
  END (* loop *);
  WriteLn; WriteLn;
  FOR Ch := 'a' TO 'z' DO
    IF Occ[Ch] <> 0 THEN
      Write (Ch); WriteString (" ");
      WriteCard (Occ[Ch], 2); WriteLn
    END (* if *)
  END (* for *)
END Example2.

```

```

MODULE Example3;
(* Check whether a word is a palindrome *)
FROM InOut IMPORT
  EOL, Done, Read, Write, WriteString, WriteLn;
VAR Word : ARRAY [1..80] OF CHAR;
  Ch      : CHAR; Len, first, second : CARDINAL;
BEGIN
  Read (Ch);
  WHILE Done DO
    Len := 1;
    WHILE Ch <> EOL DO
      Write (Ch); Word[Len] := Ch;
      INC (Len); Read (Ch)
    END (* while *);
    first := 1;
    LOOP
      second := Len - first;
      IF first >= second THEN
        WriteString (" is a palindrome");
        WriteLn; EXIT
      ELSIF (Word[first] = Word[second]) THEN
        INC (first)
      ELSE
        WriteString (" is not a palindrome");
        WriteLn; EXIT
      END (* if *)
    END (* loop *);
    Read (Ch)
  END (* while *)
END Example3.

```

```

MODULE Example4;
(* Print ASCII character set *)
FROM InOut IMPORT
  Write, WriteLn, WriteString;
VAR Ch : CHAR;
BEGIN
  FOR Ch := 0C TO 177C DO
    IF ORD (Ch) MOD 4 = 0 THEN WriteLn END;
    CASE Ch OF
      0C..37C, 177C :
        WriteString (" Control character ")
        | ' '..'/', ': '..'@',
        '[' .. ']', '{' .. '~' :
        WriteString (" Special character ")
        | '0'..'9' :
        WriteString (" Digit ")
        | 'a'..'z' :
        WriteString (" Lower case letter ")
        | 'A'..'Z' :
        WriteString (" Upper case letter ")
      END (* case *)
    END (* for *)
  END Example4.

```

```

MODULE Example5;
(* Compute the product of two matrices
   and input/output of matrices. *)
FROM InOut IMPORT
  ReadCard, WriteString, WriteLn;
FROM RealInOut IMPORT
  ReadReal, WriteReal;
CONST
  NMax = 20;
TYPE
  Matrices = ARRAY [1..NMax], [1..NMax] OF REAL;
PROCEDURE ProdMat (dim : CARDINAL; m1, m2: Matrices
  VAR res : Matrices);
VAR
  i, j, k : CARDINAL;
  el      : REAL;
BEGIN
  FOR i := 1 TO dim DO
    FOR j := 1 TO dim DO
      el := 0.0;
      FOR k := 1 TO dim DO
        el := el + m1[i,k] * m2[k,j]
      END (* for *);
      res[i,j] := el
    END (* for *)
  END (* for *)
END ProdMat;

```

```

30 PROCEDURE ReadMat (dim : CARDINAL;
31                     VAR mat : Matrices);
32   VAR
33     i, j : CARDINAL;
34   BEGIN
35     FOR i := 1 TO dim DO
36       ReadReal (mat[i,1]);
37       FOR j := 2 TO dim DO
38         WriteString (" ");
39         ReadReal (mat[i,j])
40       END (* for *);
41       WriteLn
42     END (* for *)
43   END ReadMat;
44
45 PROCEDURE WriteMat (dim : CARDINAL;
46                     mat : Matrices);
47   VAR
48     i, j : CARDINAL;
49   BEGIN
50     FOR i := 1 TO dim DO
51       WriteReal (mat[i,1], 12);
52       FOR j := 2 TO dim DO
53         WriteString (" ");
54         WriteReal (mat[i,j], 12)
55       END (* for *);
56       WriteLn
57     END (* for *)
58   END WriteMat;

```

```

59 VAR
60   Dimension : CARDINAL;
61   Mat1, Mat2, Prod : Matrices;
62
63 BEGIN
64   WriteString ("Dimension of the matrices ? ");
65   ReadCard (Dimension); WriteLn;
66   WriteString ("First matrix ?"); WriteLn;
67   ReadMat (Dimension, Mat1);
68   WriteString ("Second matrix ?"); WriteLn;
69   ReadMat (Dimension, Mat2);
70   ProdMat (Dimension, Mat1, Mat2, Prod);
71   WriteString ("Product :"); WriteLn;
72   WriteMat (Dimension, Prod)
73 END Example5.

```

Better:

```

PROCEDURE ProdMat (dim : CARDINAL;
                  VAR m1, m2, res : Matrices);

PROCEDURE WriteMat (dim : CARDINAL;
                  VAR mat : Matrices);

```

```

MODULE Example6;
(* Compute number of digits in a cardinal number *)
FROM InOut IMPORT
  ReadCard, WriteCard, WriteString, WriteLn;
PROCEDURE NrDigits (num : CARDINAL) : CARDINAL;
VAR
  nrDigits : CARDINAL;
BEGIN
  nrDigits := 0;
  REPEAT
    INC (nrDigits);
    num := num DIV 10
  UNTIL num = 0;
  RETURN nrDigits
END NrDigits;
VAR
  Num : CARDINAL;
BEGIN
  WriteString ("Give a cardinal number : ");
  ReadCard (Num); WriteLn;
  WriteString ("Number of digits = ");
  WriteCard (NrDigits (Num), 6); WriteLn;
  WriteString ("Number = ");
  WriteCard (Num, 6); WriteLn
END Example6.

```

Recursion

Fibonacci numbers

$$F_n = F_{n-1} + F_{n-2}$$


$$F_0 = F_1 = 1$$

```

PROCEDURE Fibo (n : CARDINAL) : CARDINAL;
BEGIN
  IF (n = 0) OR (n = 1) THEN
    RETURN 1
  ELSE
    RETURN Fibo (n-1) + Fibo (n-2)
  END (* if *)
END Fibo;

```

But:



Study of the evolution of a probability density distribution $\rho(x)$ over $[0, 1]$, with $\int_0^1 \rho(x) dx = 1$, under the discrete-time quadratic map

$$x_{t+1} = 4x_t(1 - x_t)$$

One obtains, $\forall t \in \mathbb{N}$

$$\rho_{t+1}(x) = \frac{1}{4\sqrt{1-x}} \cdot \left[\rho_t\left(\frac{1-\sqrt{1-x}}{2}\right) + \rho_t\left(\frac{1+\sqrt{1-x}}{2}\right) \right]$$

It can be shown that this converges to the invariant limit density

$$\lim_{t \rightarrow \infty} \rho_t(x) = \frac{1}{\pi\sqrt{x(1-x)}} \quad \forall x \in [0, 1]$$

Example of a distribution:

$$\rho_0^{(n)}(x) = \frac{(2n+1)!}{n!^2} x^n (1-x)^n$$

```

MODULE Example7;
;
FROM InOut IMPORT
  ReadCard, WriteCard, WriteString, WriteLn;
FROM RealInOut IMPORT
  WriteReal;
FROM MathLib0 IMPORT
  sqrt;
;
CONST Pi = 3.1415926536;
;
TYPE Distributions = PROCEDURE (REAL) : REAL;
;
VAR Order : CARDINAL;
    Factor : REAL;
;
PROCEDURE Rho (t : CARDINAL; x : REAL;
               rho0 : Distributions) : REAL;
  VAR
    hsqrt : REAL;
  BEGIN
    IF t = 0 THEN
      RETURN rho0 (x)
    ELSE
      hsqrt := sqrt (1.0-x) / 2.0;
      t := t-1;
      RETURN 0.125 * (Rho (t, 0.5-hsqrt, rho0)
                     + Rho (t, 0.5+hsqrt, rho0)) / hsqrt
    END (* if *)
  END Rho ;

```

```

31
32 PROCEDURE Rho0 (x : REAL) : REAL;
33 BEGIN
34     RETURN Factor * Power (Order, x * (1.0-x))
35 END Rho0;
36
37 PROCEDURE ComputeFactor (order : CARDINAL;
38                           VAR fact : REAL);
39 VAR
40     nom : CARDINAL;
41 BEGIN
42     fact := 2.0 * FLOAT(order) + 1.0;
43     nom := 2 * order;
44     WHILE order >= 1 DO
45         fact := fact * (FLOAT(nom) / FLOAT(order));
46         DEC (nom); DEC (order)
47     END (* while *)
48 END ComputeFactor;
49
50 PROCEDURE Power (n : CARDINAL; x : REAL) : REAL;
51 BEGIN
52     IF n = 0 THEN
53         RETURN 1.0
54     ELSIF n MOD 2 = 0 THEN
55         RETURN Power (n DIV 2, x * x)
56     ELSE
57         RETURN x * Power (n - 1, x)
58     END (* if *)
59 END Power;
60

```

```

61 VAR x : REAL;
62
63 BEGIN
64     WriteString ("Order of the distribution ? ");
65     ReadCard (Order); WriteLn;
66     ComputeFactor (Order, Factor);
67     x := 0.1;
68     WHILE x < 0.99 DO
69         WriteReal (Rho (5, x, Rho0), 16);
70         WriteReal (Rho (10, x, Rho0), 16);
71         WriteReal (1.0 / (Pi * sqrt (x*(1.0-x))), 16);
72         WriteLn;
73         x := x + 0.1
74     END (* while *);
75 END Example7.

```

```

1  MODULE Example8;
2  (* Compute all possible permutations
3    of a set of cardinal numbers.      *)
4  FROM InOut IMPORT
5    ReadCard, WriteCard, WriteString, WriteLn;
6
7  CONST
8    NMax = 20;
9  TYPE
10 Sequences = ARRAY [1..NMax] OF CARDINAL;
11 VAR
12   N, i : CARDINAL;
13   Perm : Sequences;
14
15 PROCEDURE Permute (VAR perm : Sequences;
16                   size, maxsize : CARDINAL);
17   VAR
18     i, tmp : CARDINAL;
19   BEGIN
20     IF size = 1 THEN
21       Print (perm, maxsize)
22     ELSE
23       FOR i := 1 TO size DO
24         tmp := perm[i]; perm[i] := perm[size];
25         perm[size] := tmp;
26         Permute (perm, size-1, maxsize);
27         tmp := perm[i]; perm[i] := perm[size];

```

```

28         perm[size] := tmp
29       END (* for *)
30     END (* if *)
31   END Permute;
32
33 PROCEDURE Print (VAR seq : Sequences;
34                 size : CARDINAL);
35 BEGIN
36   FOR i := 1 TO size DO
37     WriteCard (seq[i], 4); WriteString (" ")
38   END (* for *);
39   WriteLn
40 END Print;
41
42 BEGIN
43   WriteString (" Give a cardinal number : ");
44   ReadCard (N); WriteLn;
45   FOR i := 1 TO N DO Perm[i] := i END;
46   Permute (Perm, N, N)
47 END Example8.

```

```

10 MODULE Example8;
11 (* Compute all possible permutations
12    of a set of cardinal numbers.
13    Second version *)
14 FROM InOut IMPORT
15    ReadCard, WriteCard, WriteString, WriteLn;
16
17 CONST
18    NMax = 20;
19 TYPE
20    Sequences = ARRAY [1..NMax] OF CARDINAL;
21 VAR
22    N, i : CARDINAL;
23    Perm : Sequences;
24
25 PROCEDURE Permute (VAR perm : Sequences;
26                    maxsize : CARDINAL);
27 PROCEDURE HPermute (size : CARDINAL);
28 VAR
29    i, tmp : CARDINAL;
30 BEGIN
31     IF size = 1 THEN
32         Print (perm, maxsize)
33     ELSE
34         FOR i := 1 TO size DO
35             tmp := perm[i]; perm[i] := perm[size];
36             perm[size] := tmp;

```

```

37             HPermute (size-1);
38             tmp := perm[i]; perm[i] := perm[size];
39             perm[size] := tmp
40         END (* for *)
41     END (* if *)
42 END HPermute;
43 BEGIN
44     HPermute (maxsize)
45 END Permute;
46
47 PROCEDURE Print (VAR seq : Sequences;
48                  size : CARDINAL);
49 BEGIN
50     FOR i := 1 TO size DO
51         WriteCard (seq[i], 4); WriteString (" ")
52     END (* for *);
53     WriteLn
54 END Print;
55
56 BEGIN
57     WriteString (" Give a cardinal number : ");
58     ReadCard (N); WriteLn;
59     FOR i := 1 TO N DO Perm[i] := i END;
60     Permute (Perm, N)
61 END Example8.

```

```

. DEFINITION MODULE CmplxNum;
. (* Module for complex number arithmetic
.   and input/output *)
.
. EXPORT QUALIFIED
.   Complex,
.   Modulus, Adjoint, Add, Subtract, Multiply,
.   Divide, ReadComplex, WriteComplex;
.
. TYPE Complex = RECORD Re, Im : REAL END;
"
" PROCEDURE Modulus (c : Complex) : REAL;
" PROCEDURE Adjoint (c : Complex;
"                   VAR res : Complex);
" PROCEDURE Add (c1, c2 : Complex;
"               VAR res : Complex);
" PROCEDURE Subtract (c1, c2 : Complex;
"                   VAR res : Complex);
" PROCEDURE Multiply (c1, c2 : Complex;
"                   VAR res : Complex);
" PROCEDURE Divide (c1, c2 : Complex;
"                 VAR res : Complex);
" PROCEDURE ReadComplex (VAR c : Complex);
" PROCEDURE WriteComplex (c : Complex;
"                       width : CARDINAL);
" END CmplxNum.

```

```

" IMPLEMENTATION MODULE CmplxNum;
"
" FROM MathLib0 IMPORT sqrt;
"
" FROM InOut IMPORT
"   WriteString;
"
" FROM RealInOut IMPORT
"   WriteReal, ReadReal;
"
" PROCEDURE Modulus (c : Complex) : REAL;
" BEGIN
"   WITH c DO
"     RETURN sqrt (Re * Re + Im * Im)
"   END (* with *)
" END Modulus;
"
" PROCEDURE Adjoint (c : Complex;
"                   VAR res : Complex);
" BEGIN
"   WITH res DO
"     Re := c.Re; Im := - c.Im
"   END (* with *)
" END Adjoint;
"

```

```

12 PROCEDURE Add (c1, c2 : Complex;
13                 VAR res : Complex);
14 BEGIN
15   WITH res DO
16     Re := c1.Re + c2.Re;
17     Im := c1.Im + c2.Im
18   END (* with *)
19 END Add;
20
21 PROCEDURE Subtract (c1, c2 : Complex;
22                     VAR res : Complex);
23 BEGIN
24   WITH res DO
25     Re := c1.Re - c2.Re;
26     Im := c1.Im - c2.Im
27   END (* with *)
28 END Subtract;
29
30 PROCEDURE Multiply (c1, c2 : Complex;
31                     VAR res : Complex);
32 BEGIN
33   WITH res DO
34     Re := c1.Re * c2.Re - c1.Im * c2.Im;
35     Im := c1.Re * c2.Im + c1.Im * c2.Re
36   END (* with *)
37 END Multiply;
38
39 PROCEDURE Divide (c1, c2 : Complex;
40                  VAR res : Complex);
41   VAR sqrmod : REAL;

```

```

62 BEGIN
63   WITH c2 DO
64     sqrmod := Re*Re+Im*Im
65   END (* with *);
66   WITH res DO
67     Re := (c1.Re*c2.Re+c1.Im*c2.Im)/sqrmod;
68     Im := (c2.Re*c1.Im-c1.Re*c2.Im)/sqrmod
69   END (* with *)
70 END Divide;
71
72 PROCEDURE ReadComplex (VAR c : Complex);
73 BEGIN
74   WITH c DO
75     WriteString("("); ReadReal (Re);
76     WriteString(","); ReadReal (Im);
77     WriteString(")")
78   END (* with *)
79 END ReadComplex;
80
81 PROCEDURE WriteComplex (c : Complex; width : CARDI
82 BEGIN
83   WITH c DO
84     WriteString("("); WriteReal (Re, width);
85     WriteString(","); WriteReal (Im, width);
86     WriteString(")")
87   END (* with *)
88 END WriteComplex;
89
90 BEGIN
91 END CmplxNum.

```

```

112 MODULE CmplxTst;
113
114 FROM CmplxNum IMPORT
115     Complex,
116     Modulus, Adjoint, Add, Subtract, Multiply,
117     Divide, WriteComplex, ReadComplex;
118
119 FROM InOut IMPORT
120     WriteString, WriteLn;
121
122 FROM RealInOut IMPORT
123     WriteReal;
124
125 VAR c, c1, c2, res : Complex;
126
127 BEGIN
128     WriteString ("Give a complex number c1 : ");
129     ReadComplex (c1);
130     WriteLn;
131     WriteString ("Give a complex number c2 : ");
132     ReadComplex (c2);
133     WriteLn;
134
135     WriteString ("Modulus c1 = ");
136     WriteReal (Modulus (c1), 12); WriteLn;
137     WriteString ("Modulus c2 = ");
138     WriteReal (Modulus (c2), 12); WriteLn;
139     WriteString ("Adjoint c1 = ");
140     Adjoint (c1, res);
141     WriteComplex (res, 12); WriteLn;

```

```

142     WriteString ("Adjoint c2 = ");
143     Adjoint (c2, res);
144     WriteComplex (res, 12); WriteLn;
145
146     WriteString ("c1 + c2 = ");
147     Add (c1, c2, res);
148     WriteComplex (res, 12); WriteLn;
149     WriteString ("c1 - c2 = ");
150     Subtract (c1, c2, res);
151     WriteComplex (res, 12); WriteLn;
152     WriteString ("c1 * c2 = ");
153     Multiply (c1, c2, res);
154     WriteComplex (res, 12); WriteLn;
155     WriteString ("c1 / c2 = ");
156     Divide (c1, c2, res);
157     WriteComplex (res, 12); WriteLn;
158 END CmplxTst.

```

TEXTS AND MONOGRAPHS IN COMPUTER SCIENCE
Edited by David Gries

Niklaus Wirth

PROGRAMMING IN MODULA-2

THIRD, CORRECTED EDITION



SPRINGER-VERLAG
BERLIN HEIDELBERG NEW YORK TOKYO

functioning of the operating system and thereby also of its clients. However, Modula was conceived with the goal of serving in the construction of such operating systems as well. The inclusion of adequate device and interrupt handling facilities was therefore indispensable. Their use should nevertheless be confined to so-called stand-alone systems which do not have the support (nor the burden) of a given operating system.

Report on The Programming Language Modula-2

1. Introduction	143
2. Syntax	144
3. Vocabulary and representation	144
4. Declarations and scope rules	146
5. Constant declarations	147
6. Type declarations	147
1. Basic types	148
2. Enumerations	148
3. Subrange types	148
4. Array types	149
5. Record types	149
6. Set types	150
7. Pointer types	151
8. Procedure types	151
7. Variable declarations	151
8. Expressions	152
1. Operands	152
2. Operators	152
9. Statements	155
1. Assignments	155
2. Procedure calls	155
3. Statement sequences	156
4. If statements	156
5. Case statements	157
6. While statements	157
7. Repeat statements	157
8. For statements	158
9. Loop statements	158
10. With statements	159
11. Return and exit statements	159
10. Procedure declarations	159
1. Formal parameters	160
2. Standard procedures	162
11. Modules	163
12. System-dependent facilities	165
13. Processes	167
1. Creating a process and transfer of control	167
2. Device processes and interrupts	167
14. Compilation units	168

1. Introduction

Modula-2 grew out of a practical need for a general, efficiently implementable systems programming language for minicomputers. Its ancestors are *Pascal* and *Modula*. From the latter it has inherited the name, the important module concept, and a systematic, modern syntax, from Pascal most of the rest. This includes in particular the data structures, i.e. arrays, records, variant records, sets, and pointers. Structured statements include the familiar if, case, repeat, while, for, and with statements. Their syntax is such that every structure ends with an explicit termination symbol.

The language is essentially machine-independent, with the exception of limitations due to wordsize. This appears to be in contradiction to the notion of a system-programming language, in which it must be possible to express all operations inherent in the underlying computer. The dilemma is resolved with the aid of the *module* concept. Machine-dependent items can be introduced in specific modules, and their use can thereby effectively be confined and isolated. In particular, the language provides the possibility to relax rules about data type compatibility in these cases. In a capable system-programming language it is possible to express input/output conversion procedures, file handling routines, storage allocators, process schedulers etc. Such facilities must therefore not be included as elements of the language itself, but appear as (so-called low-level) modules which are components of most programs written. Such a collection of standard modules is therefore an essential part of a Modula-2 implementation.

The concept of processes and their synchronization with signals as included in Modula is replaced by the lower-level notion of *coroutines* in Modula-2. It is, however, possible to formulate a (standard) module that implements such processes and signals. The advantage of not including them in the language itself is that the programmer may select a process scheduling algorithm tailored to his particular needs by programming that module on his own. Such a scheduler can even be entirely omitted in simple (but frequent) cases, e.g. when concurrent processes occur as device drivers only.

A modern system programming language should in particular also facilitate the construction of large programs, possibly designed by several people. The modules written by individuals should have well-specified interfaces that can be declared independently of their actual implementations. Modula-2 supports this idea by providing separate *definition* and *implementation modules*. The former define all objects exported from the corresponding implementation module; in some cases, such as procedures and types, the definition module specifies only those parts that are relevant to the interface, i.e. to the user or client of the module.

This report is not intended as a programmer's tutorial. It is intentionally kept concise, and (we hope) clear. Its function is to serve as a reference for programmers, implementors, and manual writers, and as an arbiter, should they find disagreement.

2. Syntax

A language is an infinite set of sentences, namely the sentences well formed according to its syntax. In Modula-2, these sentences are called *compilation units*. Each unit is a finite sequence of symbols from a finite *vocabulary*. The vocabulary of Modula-2 consists of identifiers, numbers, strings, operators, and delimiters. They are called *lexical symbols* and are composed of sequences of characters. (Note the distinction between symbols and characters.)

To describe the syntax, an extended Backus-Naur Formalism called EBNF is used. Angular brackets [] denote optionality of the enclosed sentential form, and curly brackets { } denote its repetition (possibly 0 times). Syntactic entities (non-terminal symbols) are denoted by English words expressing their intuitive meaning. Symbols of the language vocabulary (terminal symbols) are strings enclosed in quote marks or words written in capital letters, so-called *reserved words*. Syntactic rules (productions) are designated by a \$ sign at the left margin of the line.

3. Vocabulary and representation

The representation of symbols in terms of characters depends on the underlying character set. The ASCII set is used in this paper, and the following lexical rules must be observed. Blanks must not occur within symbols (except in strings). Blanks and line breaks are ignored unless they are essential to separate two consecutive symbols.

1. *Identifiers* are sequences of letters and digits. The first character must be a letter.

\$ Ident = letter {letter | digit}.

Examples:

```
x scan Modula ETH GetSymbol firstLetter
```

2. *Numbers* are (unsigned) integers or real numbers. Integers are sequences of digits. If the number is followed by the letter B, it is taken as an octal number; if it is followed by the letter H, it is taken as a hexadecimal number; if it is followed by the letter C, it denotes the character with the given (octal) ordinal number (and is of type CHAR, see 6.1).

An integer i in the range $0 \leq i \leq \text{MaxInt}$ can be considered as either of type INTEGER or CARDINAL; if it is in the range $\text{MaxInt} < i \leq \text{MaxCard}$, it is of type CARDINAL. For 16-bit computers: $\text{MaxInt} = 32767$, $\text{MaxCard} = 65535$.

A real number always contains a decimal point. Optionally it may also contain a decimal scale factor. The letter E is pronounced as "ten to the power of". A real number is of type REAL.

```
number = integer | real.
integer = digit {digit} | octalDigit {octalDigit} ("B"|"C")
         digit {hexDigit} "H".
real = digit {digit} "." {digit} [ScaleFactor].
```

```
$ ScaleFactor = "E" ["+"|"-" ] digit {digit}.
$ hexDigit = digit ["A"|"B"|"C"|"D"|"E"|"F".
$ digit = octalDigit | "8"|"9".
$ octalDigit = "0"|"1"|"2"|"3"|"4"|"5"|"6"|"7".
```

Examples:

```
1980 3764B 7BCH 33C 12.3 45.67E-8
```

3. *Strings* are sequences of characters enclosed in quote marks. Both double quotes and single quotes (apostrophes) may be used as quote marks. However, the opening and closing marks must be the same character, and this character cannot occur within the string. A string must not extend over the end of a line.

\$ string = "" {character} "" | "'" {character} "'.

A string consisting of n characters is of type (see 6.4)

ARRAY [0 .. n-1] OF CHAR

Examples:

```
"MODULA" "Don't worry!" 'codeword "Barbarossa"
```

4. *Operators and delimiters* are the special characters, character pairs, or reserved words listed below. These reserved words consist exclusively of capital letters and *must not* be used in the role of identifiers. The symbols # and <> are synonyms, and so are &, AND, and ~, NOT.

+	=	AND	FOR	QUALIFIED
-	#	ARRAY	FROM	RECORD
•	<	BEGIN	IF	REPEAT
/	>	BY	IMPLEMENTATION	RETURN
:=	<>	CASE	IMPORT	SET
&	<=	CONST	IN	THEN
.	>=	DEFINITION	LOOP	TO
:	--	DIV	MOD	TYPE
:	:	DO	MODULE	UNTIL
()	ELSE	NOT	VAR
[]	ELSIF	OF	WHILE
{	}	END	OR	WITH
↑		EXIT	POINTER	
~		EXPORT	PROCEDURE	

5. *Comments* may be inserted between any two symbols in a program. They are arbitrary character sequences opened by the bracket (• and closed by •). Comments may be nested, and they do not affect the meaning of a program.

4. Declarations and scope rules

Every identifier occurring in a program must be introduced by a declaration, unless it is a standard identifier. The latter are considered to be predeclared, and they are valid in all parts of a program. For this reason they are called *pervasive*. Declarations also serve to specify certain permanent properties of an object, such as whether it is a constant, a type, a variable, a procedure, or a module.

The identifier is then used to refer to the associated object. This is possible in those parts of a program only which are within the so-called *scope* of the declaration. In general, the scope extends over the entire block (procedure or module declaration) to which the declaration belongs and to which the object is local. The scope rule is augmented by the following cases:

1. If an identifier x defined by a declaration $D1$ is used in another declaration (not statement) $D2$, then $D1$ must textually precede $D2$.
2. A type $T1$ can be used in a declaration of a pointer type T (see 6.7) which textually precedes the declaration of $T1$, if both T and $T1$ are declared in the same block. This is a relaxation of rule 1.
3. If an identifier defined in a module $M1$ is exported, the scope expands over the block which contains $M1$. If $M1$ is a compilation unit (see Ch. 14), it extends to all those units which import $M1$.
4. Field identifiers of a record declaration (see 6.5) are valid only in field designators and in with statements referring to a variable of that record type.

An identifier may be *qualified*. In this case it is prefixed by another identifier which designates the module (see Ch. 11) in which the qualified identifier is defined. The prefix and the identifier are separated by a period. Standard identifiers appear below.

\$ qualified = ident { "." ident }.

ABS	(10.2)	INCL	(10.2)
BITSET	(6.6)	INTEGER	(6.1)
BOOLEAN	(6.1)	LONGINT	(6.1)
CAP	(10.2)	LONGREAL	(6.1)
CARDINAL	(6.1)	MAX	(10.2)
CHAR	(6.1)	MIN	(10.2)
CHR	(10.2)	NIL	(6.7)
DEC	(10.2)	ODD	(10.2)
EXCL	(10.2)	ORD	(10.2)
FALSE	(6.1)	PROC	(6.8)
FLOAT	(10.2)	REAL	(6.1)
HALT	(10.2)	SIZE	(10.2)
HIGH	(10.2)	TRUE	(6.1)
INC	(10.2)	TRUNC	(10.2)
		VAL	(10.2)

5. Constant declarations

A constant declaration associates an identifier with a constant value.

```
$ ConstantDeclaration = ident "=" ConstExpression.
$ ConstExpression = expression. †
```

A constant expression is an expression which can be evaluated by a mere textual scan without actually executing the program. Its operands are constants. (see Ch. 8).

Examples of constant declarations are

```
N      = 100
limit  = 2*N-1
all    = {0 .. WordSize-1}
bound  = MAX(INTEGER) - N
```

6. Type declarations

A data type determines a set of values which variables of that type may assume, and it associates an identifier with the type. In the case of structured types, it also defines the structure of variables of this type. There are three different structures, namely arrays, records, and sets.

```
$ TypeDeclaration = ident "=" type.
$ type = SimpleType | ArrayType | RecordType | SetType |
$       PointerType | ProcedureType.
$ SimpleType = qualified { enumeration | SubrangeType.}
```

Examples:

```
Color   = (red, green, blue)
Index   = [1 .. 80]
Card    = ARRAY Index OF CHAR
Node    = RECORD key: CARDINAL;
          left, right: TreePtr
          END
Tint    = SET OF Color
TreePtr = POINTER TO Node
Function = PROCEDURE(CARDINAL): CARDINAL
```

6.1. Basic types

The following basic types are predeclared and denoted by standard identifiers:

1. INTEGER comprises the integers between MIN(INTEGER) and MAX(INTEGER).
2. CARDINAL comprises the integers between 0 and MAX(CARDINAL).
3. BOOLEAN comprises the truth values TRUE or FALSE.

4. CHAR denotes the character set provided by the used computer system.
5. REAL (and LONGREAL) denote finite sets of real numbers.
6. LONGINT comprises the integers between MIN(LONGINT) and MAX(LONGINT).

6.2. Enumerations

An enumeration is a list of identifiers that denote the values which constitute a data type. These identifiers are used as constants in the program. They, and no other values, belong to this type. The values are ordered, and the ordering relation is defined by their sequence in the enumeration. The ordinal number of the first value is 0.

```
$ enumeration = "(" IdentList ")".
$ IdentList = ident {"." ident}.
```

Examples of enumerations:

```
(red, green, blue)
(club, diamond, heart, spade)
(Monday, Tuesday, Wednesday, Thursday, Friday, Saturday, Sunday)
```

6.3. Subrange types

A type T may be defined as a subrange of another, basic or enumeration type T1 (except REAL) by specification of the least and the highest value in the subrange.

```
$ SubrangeType = [ident] "[" ConstExpression ".." ConstExpression "]". ↑
```

The first constant specifies the lower bound, and must not be greater than the upper bound. The type T1 of the bounds is called the *base type* of T, and all operators applicable to operands of type T1 are also applicable to operands of type T. However, a value to be assigned to a variable of a subrange type must lie within the specified interval. The base type can be specified by an identifier preceding the bounds. If it is omitted, and if the lower bound is a non-negative integer, the base type of the subrange is taken to be CARDINAL; if it is a negative integer, it is INTEGER.

A type T1 is said to be *compatible* with a type T0, if it is declared either as T1 = T0 or as a subrange of T0, or if T0 is a subrange of T1, or if T0 and T1 are both subranges of the same (base) type.

Examples of subrange types:

```
[0 .. N-1]
["A" .. "Z"]
[Monday .. Friday]
```

6.4. Array types

An array is a structure consisting of a fixed number of components which are all of the same type, called the *component type*. The elements of the array are designated by indices, values

belonging to the *index type*. The array type declaration specifies the component type as well as the index type. The latter must be an enumeration, a subrange type, or one of the basic types BOOLEAN or CHAR.

```
$ ArrayType = ARRAY SimpleType {"." SimpleType} OF type.
```

A declaration of the form

```
ARRAY T1, T2, ..., Tn OF T
```

with n index types T1 ... Tn must be understood as an abbreviation for the declaration

```
ARRAY T1 OF
  ARRAY T2 OF
    ...
    ARRAY Tn OF T
```

Examples of array types:

```
ARRAY [0 .. N-1] OF CARDINAL
ARRAY [1 .. 10], [1..20] OF [0 .. 99]
ARRAY [-10 .. +10] OF BOOLEAN
ARRAY WeekDay OF Color
ARRAY Color OF WeekDay
```

6.5. Record types

A record type is a structure consisting of a fixed number of components of possibly different types. The record type declaration specifies for each component, called *field*, its type and an identifier which denotes the field. The scope of these field identifiers is the record definition itself, and they are also accessible within field designators (see 8.1) referring to components of record variables, and within with statements.

A record type may have several variant sections, in which case the first field of the section is called the *tag field*. Its value indicates which variant is assumed by the section. Individual variant structures are identified by *case labels*. These labels are constants of the type indicated by the tag field.

```
$ RecordType = RECORD FieldListSequence END.
$ FieldListSequence = FieldList {"." FieldList}.
$ FieldList = [IdentList ":" type |
$   CASE [ident] ":" qualified OF variant {"|" variant} ↑
$   [ELSE FieldListSequence] END.
$ variant = [CaseLabelList ":" FieldListSequence]. ↑
$ CaseLabelList = CaseLabels {"." CaseLabels}.
$ CaseLabels = ConstExpression [".." ConstExpression].
```

Examples of record types:

```
RECORD day: [1 .. 31];
  month: [1 .. 12];
```

```

    year: [0 .. 2000]
END
RECORD
    name,firstname: ARRAY [0 .. 9] OF CHAR;
    age: [0 .. 99];
    salary: REAL
END
RECORD x, y: T0;
    CASE tag0: Color OF
        red: a: Tr1; b: Tr2 |
        green: c: Tg1; d: Tg2 |
        blue: e: Tb1; f: Tb2
    END;
    z: T0;
    CASE tag1: BOOLEAN OF
        TRUE: u,v: INTEGER |
        FALSE: r,s: CARDINAL
    END
END

```

The example above contains two variant sections. The variant of the first section is indicated by the value of the tag field tag0, the one of the second section by the tag field tag1.

6.6. Set types

A set type defined as SET OF T comprises all sets of values of its base type T. This must be a subrange of the integers between 0 and N-1, or a (subrange of an) enumeration type with at most N values, where N is a small constant determined by the implementation, usually the computer's wordsize or a small multiple thereof.

\$ SetType = SET OF SimpleType.

The standard type BITSET is defined as follows, where W is a constant defined by the implementation, usually the word size of the computer.

BITSET = SET OF [0 .. W-1]

6.7. Pointer types

Variables of a pointer type P assume as values pointers to variables of another type T. The pointer type P is said to be *bound* to T. A pointer value is generated by a call to an allocation procedure in a storage management module.

\$ PointerType = POINTER TO type.

Besides such pointer values, a pointer variable may assume the value NIL, which can be thought as pointing to no variable at all.

6.8. Procedure types

Variables of a procedure type T may assume as their value a procedure P. The (types of the) formal parameters of P must be the same as those indicated in the formal type list of T. The same holds for the result type in the case of a function procedure.

Restriction: P must not be declared local to another procedure, and neither can it be a standard procedure.

```

$ ProcedureType = PROCEDURE [FormalTypeList].
$ FormalTypeList = "(" [[VAR] FormalType
$   {"," [VAR] FormalType} "]" [":" qualident].

```

The standard type PROC denotes a parameterless procedure:

PROC = PROCEDURE

7. Variable declarations

Variable declarations serve to introduce variables and associate them with a unique identifier and a fixed data type and structure. Variables whose identifiers appear in the same list all obtain the same type.

\$ VariableDeclaration = IdentList ":" type.

The data type determines the set of values that a variable may assume and the operators that are applicable; it also defines the structure of the variable.

Examples of variable declarations (refer to examples in Ch. 6):

```

l, j: CARDINAL
k: INTEGER
p, q: BOOLEAN
s: BITSET
F: Function
a: ARRAY Index OF CARDINAL
w: ARRAY [0 .. 7] OF
    RECORD ch : CHAR;
        count : CARDINAL
    END

```

8. Expressions

Expressions are constructs denoting rules of computation for obtaining values of variables and generating new values by the application of operators. Expressions consist of operands and operators. Parentheses may be used to express specific associations of operators and operands.

8.1. Operands

With the exception of literal constants, i.e. numbers, character strings, and sets (see Ch. 5), operands are denoted by *designators*. A designator consists of an identifier referring to the constant, variable, or procedure to be designated. This identifier may possibly be qualified by module identifiers (see Ch. 4 and 11), and it may be followed by selectors, if the designated object is an element of a structure. If the structure is an array A, then the designator A[E] denotes that component of A whose index is the current value of the expression E. The index type of A must be *assignment compatible* with the type of E (see 9.1). A designator of the form

A[E1, E2, ..., En] stands for A[E1][E2]...[En].

If the structure is a record R, then the designator R.f denotes the record field f of R. The designator P+ denotes the variable which is referenced by the pointer P.

\$ designator = qualident { "." ident | "[" ExpList "]" | "+" }.

\$ ExpList = expression { "," expression }.

If the designated object is a variable, then the designator refers to the variable's current value. If the object is a function procedure, a designator without parameter list refers to that procedure. If it is followed by a (possibly empty) parameter list, the designator implies an activation of the procedure and stands for the value resulting from its execution, i.e. for the "returned" value. The (types of these) actual parameters must correspond to the formal parameters as specified in the procedure's declaration (see Ch. 10).

Examples of designators (see examples in Ch. 7):

k	(INTEGER)
a[i]	(CARDINAL)
w[3].ch	(CHAR)
tr.key	(CARDINAL)
tr.left.tr.right	(TreePtr)

8.2. Operators

The syntax of expressions specifies operator precedences according to four classes of operators. The operator NOT has the highest precedence, followed by the so-called multiplying operators, then the so-called adding operators, and finally, with lowest precedence, the relational operators. Sequences of operators of the same precedence are executed from left to right.

\$ expression = SimpleExpression [relation SimpleExpression].
 \$ relation = "=" | "<" | "<=" | ">" | ">=" | "IN".
 \$ SimpleExpression = ["+" | "-"] term {AddOperator term}.
 \$ AddOperator = "+" | "-" | OR.
 \$ term = factor {MulOperator factor}.
 \$ MulOperator = "*" | "/" | DIV | MOD | AND.
 \$ factor = number | string | set | designator [ActualParameters] |

\$ "(" expression ")" | NOT factor.
 \$ set = [qualident] "[" [element { "," element }] "]" .
 \$ element = expression [".." expression] .
 \$ ActualParameters = "(" [ExpList] ")" .

The available operators are listed in the following tables. In some instances, several different operations are designated by the same operator symbol. In these cases, the actual operation is identified by the types of the operands.

8.2.1. Arithmetic operators

symbol	operation
+	addition
-	subtraction
*	multiplication
/	real division
DIV	integer division
MOD	modulus

These operators (except /) apply to operands of type INTEGER, CARDINAL, or subranges thereof. Both operands must be either of type CARDINAL or a subrange with base type CARDINAL, in which case the result is of type CARDINAL, or they must both be of type INTEGER or a subrange with base type INTEGER, in which case the result is of type INTEGER.

The operators +, -, and * also apply to operands of type REAL. In this case, both operands must be of type REAL, and the result is then also of type REAL. The division operator / applies to REAL operands only. When used as operators with a single operand only, - denotes sign inversion and + denotes the identity operation. Sign inversion applies to operands of type INTEGER or REAL. The operations DIV and MOD are defined by the following rules:

$x \text{ DIV } y$ is equal to the truncated quotient of x/y
 $x \text{ MOD } y$ is equal to the remainder of the division $x \text{ DIV } y$ (for $y > 0$)
 $x = (x \text{ DIV } y) * y + (x \text{ MOD } y)$

8.2.2. Logical operators

symbol	operation
OR	logical conjunction
AND	logical disjunction
NOT	negation

These operators apply to BOOLEAN operands and yield a BOOLEAN result.

p OR q means "if p then TRUE, otherwise q"
 p AND q means "if p then q, otherwise FALSE"

8.2.3. Set operators

symbol	operation
+	set union
-	set difference
•	set intersection
/	symmetric set difference

These operations apply to operands of any set type and yield a result of the same type.

$x \text{ IN } (s1 + s2)$	iff	$(x \text{ IN } s1) \text{ OR } (x \text{ IN } s2)$
$x \text{ IN } (s1 - s2)$	iff	$(x \text{ IN } s1) \text{ AND NOT } (x \text{ IN } s2)$
$x \text{ IN } (s1 \bullet s2)$	iff	$(x \text{ IN } s1) \text{ AND } (x \text{ IN } s2)$
$x \text{ IN } (s1 / s2)$	iff	$(x \text{ IN } s1) \# (x \text{ IN } s2)$

8.2.4. Relations

Relations yield a Boolean result. The ordering relations apply to the basic types INTEGER, CARDINAL, BOOLEAN, CHAR, REAL, to enumerations, and to subrange types.

symbol	relation
=	equal
#	unequal
<	less
<=	less or equal (set inclusion)
>	greater
>=	greater or equal (set inclusion)
IN	contained in (set membership)

The relations = and # also apply to sets and pointers. If applied to sets, <= and >= denote (improper) inclusion. The relation IN denotes set membership. In an expression of the form $x \text{ IN } s$, the expression s must be of type SET OF T, where T is (compatible with) the type of x .

Examples of expressions (refer to examples in Ch. 7):

1980	(CARDINAL)
k DIV 3	(INTEGER)
NOT p OR q	(BOOLEAN)
(i+j) * (i-j)	(CARDINAL)
s - {8,9,13}	(BITSET)
a[i] + a[j]	(CARDINAL)
a[i+j] * a[i-j]	(CARDINAL)
(0 < k) & (k < 100)	(BOOLEAN)
tr.key = 0	(BOOLEAN)
{13..15} <= s	(BOOLEAN)
i IN {0, 5 .. 8, 15}	(BOOLEAN)

9. Statements

Statements denote actions. There are elementary and structured statements. Elementary statements are not composed of any parts that are themselves statements. They are the assignment, the procedure call, and the return and exit statements. Structured statements are composed of parts that are themselves statements. These are used to express sequencing, and conditional, selective, and repetitive execution.

```
$ statement = {assignment | ProcedureCall |
               IfStatement | CaseStatement | WhileStatement |
               RepeatStatement | LoopStatement | ForStatement |
               WithStatement | EXIT | RETURN [expression] }.
```

A statement may also be empty, in which case it denotes no action. The empty statement is included in order to relax punctuation rules in statement sequences.

9.1. Assignments

The assignment serves to replace the current value of a variable by a new value indicated by an expression. The assignment operator is written as ":=" and pronounced as "becomes".

```
$ assignment = designator ":=" expression.
```

The designator to the left of the assignment operator denotes a variable. After an assignment is executed, the variable has the value obtained by evaluating the expression. The old value is lost (overwritten). The type of the variable must be assignment compatible with the type of the expression. Operand types are said to be *assignment compatible*, if either they are compatible or both are INTEGER or CARDINAL or subranges with base types INTEGER or CARDINAL.

A string of length $n1$ can be assigned to a string variable of length $n2 > n1$. In this case, the string value is extended with a null character (0C). A string of length 1 is compatible with the type CHAR. †

Examples of assignments:

```
i := k
p := i = j
j := log2(i+j)
F := log2
s := {2,3,5,7,11,13}
a[i] := (i+j) * (i-j)
tr.key := 1
w[i+1].ch := "A"
```

9.2. Procedure calls

A procedure call serves to activate a procedure. The procedure call may contain a list of

actual parameters which are substituted in place of their corresponding formal parameters defined in the procedure declaration (see Ch. 10). The correspondence is established by the positions of the parameters in the lists of actual and formal parameters respectively. There exist two kinds of parameters: *variable* and *value parameters*.

In the case of variable parameters, the actual parameter must be a designator denoting a variable. If it designates a component of a structured variable, the selector is evaluated when the formal/actual parameter substitution takes place, i.e. before the execution of the procedure. If the parameter is a value parameter, the corresponding actual parameter must be an expression. This expression is evaluated prior to the procedure activation, and the resulting value is assigned to the formal parameter which now constitutes a local variable. The types of corresponding actual and formal parameters must be identical in the case of variable parameters, or assignment compatible in the case of value parameters.

\$ ProcedureCall = designator [ActualParameters].

Examples of procedure calls:

```
Read(i)           (see Ch. 10)
Write(j*2+1.6)
INC(a[i])
```

9.3. Statement sequences

Statement sequences denote the sequence of actions specified by the component statements which are separated by semicolons.

\$ StatementSequence = statement {";" statement}.

9.4. If statements

```
$ IfStatement = IF expression THEN StatementSequence
$ {ELSIF expression THEN StatementSequence}
$ [ELSE StatementSequence] END.
```

The expressions following the symbols IF and ELSIF are of type BOOLEAN. They are evaluated in the sequence of their occurrence, until one yields the value TRUE. Then its associated statement sequence is executed. If an ELSE clause is present, its associated statement sequence is executed if and only if all Boolean expressions yielded the value FALSE.

Example:

```
IF (ch >= "A") & (ch <= "Z") THEN ReadIdentifier
ELSIF (ch >= "0") & (ch <= "9") THEN ReadNumber
ELSIF ch = "" THEN ReadString("")
ELSIF ch = " " THEN ReadString(" ")
ELSE SpecialCharacter
END
```

9.5. Case statements

Case statements specify the selection and execution of a statement sequence according to the value of an expression. First the case expression is evaluated, then the statement sequence is executed whose case label list contains the obtained value. The type of the case expression must be a basic type (except REAL), an enumeration type, or a subrange type, and all labels must be compatible with that type. Case labels are constants, and no value must occur more than once. If the value of the expression does not occur as a label of any case, the statement sequence following the symbol ELSE is selected.

```
$ CaseStatement = CASE expression OF case {"|" case}
$ [ELSE StatementSequence] END.
$ case = [CaseLabelList ":" StatementSequence]. †
```

Example:

```
CASE 1 OF
0: p := p OR q; x := x + y |
1: p := p OR q; x := x - y |
2: p := p AND q; x := x * y
END
```

9.6. While statements

While statements specify the repeated execution of a statement sequence depending on the value of a Boolean expression. The expression is evaluated before each subsequent execution of the statement sequence. The repetition stops as soon as this evaluation yields the value FALSE.

\$ WhileStatement = WHILE expression DO StatementSequence END.

Examples:

```
WHILE j > 0 DO
j := j DIV 2; i := i + 1
END

WHILE i # j DO
IF i > j THEN i := i - j
ELSE j := j - i
END
END

WHILE (t # NIL) & (t.key # 1) DO
t := t.left
END
```

9.7. Repeat statements

Repeat statements specify the repeated execution of a statement sequence depending on the

value of a Boolean expression. The expression is evaluated after each execution of the statement sequence, and the repetition stops as soon as it yields the value TRUE. Hence, the statement sequence is executed at least once.

\$ RepeatStatement = REPEAT StatementSequence UNTIL expression.

Example:

```
REPEAT k := 1 MOD j; i := j; j := k
UNTIL j = 0
```

9.8. For statements

The for statement indicates that a statement sequence is to be repeatedly executed while a progression of values is assigned to a variable. This variable is called the *control variable* of the for statement. It cannot be a component of a structured variable, it cannot be imported, nor can it be a parameter. Its value should not be changed by the statement sequence.

\$ ForStatement = FOR Ident ":" = " expression TO expression
[BY ConstExpression] DO StatementSequence END.

The for statement

```
FOR v := A TO B BY C DO SS END
```

expresses repeated execution of the statement sequence SS with v successively assuming the values A, A+C, A+2C, ..., A+nC, where A+nC is the last term not exceeding B. v is called the control variable, A the starting value, B the limit, and C the increment. A and B must be compatible (†) with v; C must be a constant of type INTEGER or CARDINAL. If no increment is specified, it is assumed to be 1.

Examples:

```
FOR i := 1 TO 80 DO j := j + a[i] END
FOR i := 80 TO 2 BY -1 DO a[i] := a[i-1] END
```

9.9. Loop statements

A loop statement specifies the repeated execution of a statement sequence. It is terminated by the execution of any exit statement within that sequence.

\$ LoopStatement = LOOP StatementSequence END.

Example:

```
LOOP
IF t1.key > x THEN t2 := t1.left; p := TRUE
ELSE t2 := t1.right; p := FALSE
END;
IF t2 = NIL THEN
EXIT
END;
```

```
t1 := t2
END
```

While, repeat, and for statements can be expressed by loop statements containing a single exit statement. Their use is recommended as they characterize the most frequently occurring situations where termination depends either on a single condition at either the beginning or end of the repeated statement sequence, or on reaching the limit of an arithmetic progression. The loop statement is, however, necessary to express the continuous repetition of cyclic processes, where no termination is specified. It is also useful to express situations exemplified above. Exit statements are contextually, although not syntactically bound to the loop statement which contains them.

9.10. With statements

The with statement specifies a record variable and a statement sequence. In these statements the qualification of field identifiers may be omitted, if they are to refer to the variable specified in the with clause. If the designator denotes a component of a structured variable, the selector is evaluated once (before the statement sequence). The with statement opens a new scope.

\$ WithStatement = WITH designator DO StatementSequence END.

Example:

```
WITH tt DO
key := 0; left := NIL; right := NIL
END
```

9.11. Return and exit statements

A return statement consists of the symbol RETURN, possibly followed by an expression. It indicates the termination of a procedure (or a module body), and the expression specifies the value returned as result of a function procedure. Its type must be assignment compatible with the result type specified in the procedure heading (see Ch. 10).

Function procedures require the presence of a return statement indicating the result value. There may be several, although only one will be executed. In proper procedures, a return statement is implied by the end of the procedure body. An explicit return statement therefore appears as an additional, probably exceptional termination point.

An exit statement consists of the symbol EXIT, and it specifies termination of the enclosing loop statement and continuation with the statement following that loop statement (see 9.9).

10. Procedure declarations

Procedure declarations consist of a *procedure heading* and a block which is said to be the *procedure body*. The heading specifies the procedure identifier and the *formal parameters*. The block contains declarations and statements. The procedure identifier is repeated at the end of the procedure declaration.

There are two kinds of procedures, namely *proper procedures* and *function procedures*. The latter are activated by a function designator as a constituent of an expression, and yield a result that is an operand in the expression. Proper procedures are activated by a procedure call. The function procedure is distinguished in the declaration by indication of the type of its result following the parameter list. Its body must contain a RETURN statement which defines the result of the function procedure.

All constants, variables, types, modules and procedures declared within the block that constitutes the procedure body are *local* to the procedure. The values of local variables, including those defined within a local module, are undefined upon entry to the procedure. Since procedures may be declared as local objects too, procedure declarations may be nested. Every object is said to be declared at a certain *level* of nesting. If it is declared local to a procedure at level k , it has itself level $k+1$. Objects declared in the module that constitutes a compilation unit (see Ch. 14) are defined to be at level 0.

In addition to its formal parameters and local objects, also the objects declared in the environment of the procedure are known and accessible in the procedure (with the exception of those objects that have the same name as objects declared locally).

The use of the procedure identifier in a call within its declaration implies recursive activation of the procedure.

```
$ ProcedureDeclaration = ProcedureHeading ":" block Ident.
$ ProcedureHeading = PROCEDURE Ident [FormalParameters].
$ block = {declaration} [BEGIN StatementSequence] END.
$ declaration = CONST {ConstantDeclaration ":"} |
$   TYPE {TypeDeclaration ":"} |
$   VAR {VariableDeclaration ":"} |
$   ProcedureDeclaration ":" | ModuleDeclaration ":".
```

10.1. Formal parameters

Formal parameters are identifiers which denote actual parameters specified in the procedure call. The correspondence between formal and actual parameters is established when the procedure is called. There are two kinds of parameters, namely *value* and *variable parameters*. The kind is indicated in the formal parameter list. Value parameters stand for local variables to which the result of the evaluation of the corresponding actual parameter is assigned as initial value. Variable parameters correspond to actual parameters that are variables, and they stand for these variables. Variable parameters are indicated by the symbol VAR, value parameters by the absence of the symbol VAR.

Formal parameters are local to the procedure, i.e. their scope is the program text which constitutes the procedure declaration.

```
$ FormalParameters =
$   "(" [FPSection {" ":" FPSection"}] ")" [" ":" qualident].
$ FPSection = [VAR] IdentList ":" FormalType.
$ FormalType = [ARRAY OF] qualident.
```

The type of each formal parameter is specified in the parameter list. In the case of variable parameters it must be identical (†) with its corresponding actual parameter (see 9.2, and 12. for exceptions); in the case of value parameters the formal type must be assignment compatible with the actual type (see 9.1). If the parameter is an array, the form

ARRAY OF T

may be used, where the specification of the actual index bounds is omitted. The parameter is then said to be an *open array parameter*. T must be the same as the element type of the actual array, and the index range is mapped onto the integers 0 to $N-1$, where N is the number of elements. The formal array can be accessed elementwise only, or it may occur as actual parameter whose formal parameter is without specified index bounds. A function procedure without parameters has an empty parameter list. It must be called by a function designator whose actual parameter list is empty too.

Restriction: If a formal parameter specifies a procedure type, then the corresponding actual parameter must be either a procedure declared at level 0 or a variable (or parameter) of that procedure type. It cannot be a standard procedure.

Examples of procedure declarations:

```
PROCEDURE Read(VAR x: CARDINAL);
  VAR i: CARDINAL; ch: CHAR;
  BEGIN i := 0;
    REPEAT ReadChar(ch)
      UNTIL (ch >= "0") & (ch <= "9");
    REPEAT i := 10*i + (ORD(ch)-ORD("0"));
      ReadChar(ch)
    UNTIL (ch < "0") OR (ch > "9");
    x := i
  END Read
```

```
PROCEDURE Write(x,n: CARDINAL);
  VAR i: CARDINAL;
  buf: ARRAY [1..10] OF CARDINAL;
  BEGIN i := 0;
    REPEAT INC(i); buf[i] := x MOD 10; x := x DIV 10
      UNTIL x = 0;
    WHILE n > i DO
      WriteChar(" "); DEC(n)
    END ;
    REPEAT WriteChar(CHR(buf[i] + ORD("0")));
      DEC(i)
    UNTIL i = 0;
  END Write
```

```
PROCEDURE log2(x: CARDINAL): CARDINAL;
  VAR y: CARDINAL; (*assume x>0*)
  BEGIN x := x-1; y := 0;
```

```

WHILE x > 0 DO
  x := x DIV 2; y := y + 1
END;
RETURN y
END log2

```

10.2. Standard procedures

Standard procedures are predefined. Some are *generic* procedures that cannot be explicitly declared, i.e. they apply to classes of operand types or have several possible parameter list forms. Standard procedures are

ABS(x)	absolute value; result type = argument type.
CAP(ch)	If ch is a lower case letter, the corresponding capital letter; If ch is a capital letter, the same letter.
CHR(x)	the character with ordinal number x. $\text{CHR}(x) = \text{VAL}(\text{CHAR}, x)$
FLOAT(x)	x of type CARDINAL represented as a value of type REAL.
HIGH(a)	high index bound of array a.
MAX(T)	the maximum value of type T. \uparrow
MIN(T)	the minimum value of type T. \uparrow
ODD(x)	$x \bmod 2 \neq 0$.
ORD(x)	ordinal number (of type CARDINAL) of x in the set of values defined by type T of x. T is any enumeration type, CHAR, INTEGER, or CARDINAL.
SIZE(T)	the number of storage units required by a variable of type T, or the number of storage units required by the variable T. \uparrow
TRUNC(x)	real number x truncated to its integral part (of type CARDINAL).
VAL(T, x)	the value with ordinal number x and with type T. T is any enumeration type, or CHAR, INTEGER, or CARDINAL. $\text{VAL}(T, \text{ORD}(x)) = x$, if x of type T.
DEC(x)	$x := x - 1$
DEC(x, n)	$x := x - n$
EXCL(s, i)	$s := s - \{i\}$
HALT	terminate program execution
INC(x)	$x := x + 1$
INC(x, n)	$x := x + n$
INCL(s, i)	$s := s + \{i\}$

The procedures INC and DEC also apply to operands x of enumeration types and of type CHAR. In these cases they replace x by its (n-th) successor or predecessor.

11. Modules

A module constitutes a collection of declarations and a sequence of statements. They are enclosed in the brackets MODULE and END. The module heading contains the module identifier, and possibly a number of *import lists* and an *export list*. The former specify all identifiers of objects that are declared outside but used within the module and therefore have to be imported. The export-list specifies all identifiers of objects declared within the module and used outside. Hence, a module constitutes a wall around its local objects whose transparency is strictly under control of the programmer.

Objects local to a module are said to be at the same scope level as the module. They can be considered as being local to the procedure enclosing the module but residing within a more restricted scope.

```

$ ModuleDeclaration =
$   MODULE Ident [priority] ":" {import} {export} block Ident.
$   priority = "[" ConstExpression "]".
$   export = EXPORT [QUALIFIED] IdentList ":".
$   import = [FROM Ident] IMPORT IdentList ":".

```

The module identifier is repeated at the end of the declaration.

The statement sequence that constitutes the *module body* is executed when the procedure to which the module is local is called. If several modules are declared, then these bodies are executed in the sequence in which the modules occur. These bodies serve to initialize local variables and must be considered as prefixes to the enclosing procedure's statement part.

If an identifier occurs in the import (export) list, then the denoted object may be used inside (outside) the module as if the module brackets did not exist. If, however, the symbol EXPORT is followed by the symbol QUALIFIED, then the listed identifiers must be prefixed with the module's identifier when used outside the module. This case is called *qualified export*, and is used when modules are designed which are to be used in coexistence with other modules not known a priori. Qualified export serves to avoid clashes of identical identifiers exported from different modules (and presumably denoting different objects).

A module may feature several import lists which may be prefixed with the symbol FROM and a module identifier. The FROM clause has the effect of unqualifying the imported identifiers. Hence they may be used within the module as if they had been exported in normal, i.e. non-qualified mode.

If a record type is exported, all its field identifiers are exported too. The same holds for the constant identifiers in the case of an enumeration type.

Examples of module declarations:

The following module serves to scan a text and to copy it into an output character sequence. Input is obtained characterwise by a procedure Inchr and delivered by a procedure Outchr. The characters are given in the ASCII code; control characters are ignored, with the

exception of LF (line feed) and FS (file separator). They are both translated into a blank and cause the Boolean variables `coln` (end of line) and `eof` (end of file) to be set respectively. FS is assumed to be preceded by LF.

```

MODULE LineInput;
IMPORT Inchr, outchr;
EXPORT read, NewLine, NewFile, coln, eof, lno;
CONST LF = 12C; CR = 15C; FS = 34C;

VAR lno: CARDINAL; (*line number*)
    ch: CHAR; (*last character read*)
    eof, coln: BOOLEAN;

PROCEDURE NewFile;
BEGIN
    IF NOT eof THEN
        REPEAT Inchr(ch) UNTIL ch = FS;
    END;
    eof := FALSE; coln := FALSE; lno := 0
END NewFile;

PROCEDURE NewLine;
BEGIN
    IF NOT coln THEN
        REPEAT Inchr(ch) UNTIL ch = LF;
        outchr(CR); outchr(LF)
    END;
    coln := FALSE; INC(lno)
END NewLine;

PROCEDURE read(VAR x: CHAR);
BEGIN (*assume NOT coln AND NOT eof*)
    LOOP Inchr(ch); outchr(ch);
        IF ch >= " " THEN
            x := ch; EXIT
        ELSIF ch = LF THEN
            x := " "; coln := TRUE; EXIT
        ELSIF ch = FS THEN
            x := " "; coln := TRUE; eof := TRUE; EXIT
        END
    END
END read;

BEGIN eof := TRUE; coln := TRUE
END LineInput

```

The next example is a module which operates a disk track reservation table, and protects it from unauthorized access. A function procedure `NewTrack` yields the number of a free

track which is becoming reserved. Tracks can be released by calling procedure `ReturnTrack`.

```

MODULE TrackReservation;
EXPORT NewTrack, ReturnTrack;
CONST ntr = 1024; (* no. of tracks *)
    w = 16; (* word size *)
    m = ntr DIV w;

VAR l: CARDINAL;
    free: ARRAY [0..m-1] OF BITSET;

PROCEDURE NewTrack(): INTEGER;
(*reserves a new track and yields its index as result,
  if a free track is found, and -1 otherwise*)
VAR i, j: CARDINAL; found: BOOLEAN;
BEGIN found := FALSE; l := m;
    REPEAT DEC(i); j := w;
        REPEAT DEC(j);
            IF j IN free[i] THEN found := TRUE END
        UNTIL found OR (j = 0)
        UNTIL found OR (i = 0);
    IF found THEN EXCL(free[i], j); RETURN (w + j)
    ELSE RETURN -1
    END
END NewTrack;

PROCEDURE ReturnTrack(k: CARDINAL);
BEGIN (*assume 0 <= k < ntr *)
    INCL(free[k DIV w], k MOD w)
END ReturnTrack;

BEGIN (*mark all tracks free*)
    FOR l := 0 TO m-1 DO free[l] := {0..w-1} END
END TrackReservation

```

12. System-dependent facilities

Modula-2 offers certain facilities that are necessary to program *low-level* operations referring directly to objects particular of a given computer and/or implementation. These include for example facilities for accessing devices that are controlled by the computer, and facilities to break the data type compatibility rules otherwise imposed by the language definition. Such facilities are to be used with utmost care, and it is strongly recommended to restrict their use to specific modules (called low-level modules). Most of them appear in the form of data types and procedures imported from the standard module `SYSTEM`. A low-level module is therefore explicitly characterized by the identifier `SYSTEM` appearing in its import list.

Note: Because the objects imported from SYSTEM obey special rules, this module must be known to the compiler. It is therefore called a pseudo-module and need not be supplied as a separate definition module (see Ch. 14).

The facilities exported from the module SYSTEM are specified by individual implementations. Normally, the types WORD and ADDRESS, and the procedures ADR, TSIZE, NEWPROCESS, TRANSFER, are among them (see also Ch. 13).

The type WORD represents an individually accessible storage unit. No operation except assignment is defined on this type. However, if a formal parameter of a procedure is of type WORD, the corresponding actual parameter may be of any type that uses one storage word in the given implementation. If a formal parameter has the type ARRAY OF WORD, its corresponding actual parameter may be of any type; in particular it may be a record type to be interpreted as an array of words.

The type ADDRESS is defined as

ADDRESS = POINTER TO WORD

It is compatible with all pointer types, and also with the type CARDINAL. Therefore, all operators for integer arithmetic apply to operands of this type. Hence, the type ADDRESS can be used to perform address computations and to export the results as pointers. If a formal parameter is of type ADDRESS, the corresponding actual parameter may be of any pointer type, even if the formal parameter is a VAR parameter. The following example of a primitive storage allocator demonstrates a typical usage of the type ADDRESS.

```
MODULE Storage;
  FROM SYSTEM IMPORT ADDRESS;
  EXPORT Allocate;

  VAR lastused: ADDRESS;

  PROCEDURE Allocate (VAR a: ADDRESS; n: CARDINAL);
    BEGIN a := lastused; lastused := lastused + n
  END Allocate;

  BEGIN lastused := 0
END Storage
```

The function ADR(x) denotes the storage address of the variable x and is of type ADDRESS. TSIZE(T) is the number of storage units assigned to any variable of type T. TSIZE is of an arithmetic type depending on the implementation. (†)

Examples:

ADR(lastused) TSIZE(Node)

Besides those exported from the pseudo-module SYSTEM, there are two other facilities whose characteristics are system-dependent. The first is the possibility to use a type identifier T as a name denoting the *type transfer function* from the type of the operand to the type T. Evidently, such functions are data representation dependent, and they involve no explicit conversion instructions.

The second, non-standard facility may be provided in variable declarations. It allows to specify the absolute address of a variable and to override the allocation scheme of a compiler. This facility is intended for access to storage locations with specific purpose and fixed address, such as e.g. device registers on computers with "memory-mapped I/O". This address is specified as a constant integer expression enclosed in brackets immediately following the identifier in the variable declaration. The choice of an appropriate data type is left to the programmer.

13. Processes

Modula-2 is designed primarily for implementation on a conventional single-processor computer. For multiprogramming it offers only some basic facilities which allow the specification of quasi-concurrent processes and of genuine concurrency for peripheral devices. The word *process* is here used with the meaning of *coroutine*. Coroutines are processes that are executed by a (single) processor one at a time.

13.1. Creating a process and transfer of control

A new process is created by a call to

```
PROCEDURE NEWPROCESS(P: PROC;
  A: ADDRESS; n: CARDINAL; VAR p1: ADDRESS) †
```

P denotes the procedure which constitutes the process,
A is the base address of the process' workspace,
n is the size of this workspace,
p1 is the result parameter.

A new process with P as program and A as workspace of size n is assigned to p1. This process is allocated, but not activated. P must be a parameterless procedure declared at level 0.

A transfer of control between two processes is specified by a call to

```
PROCEDURE TRANSFER(VAR p1, p2: ADDRESS) †
```

This call suspends the current process, assigns it to p1, and resumes the process designated by p2. Evidently, p2 must have been assigned a process by an earlier call to either NEWPROCESS or TRANSFER. Both procedures must be imported. A program terminates, when control reaches the end of a procedure which is the body of a process.

Note: assignment to p1 occurs after identification of the new process p2; hence, the actual parameters may be identical.

13.2. Device processes and interrupts

If a process contains an operation of a peripheral device, then the processor may be transferred to another process after the operation of the device has been initiated, thereby leading to a concurrent execution of that other process with the *device process*. Usually, termination of the device's operation is signalled by an interrupt of the main processor. In

terms of Modula-2, an interrupt is a transfer operation. This interrupt transfer is (in Modula-2 implemented on the PDP-11) preprogrammed by and combined with the transfer after device initiation. This combination is expressed by a call to

```
PROCEDURE IOTRANSFER(VAR p1, p2: ADDRESS; va: CARDINAL) ↑
```

In analogy to TRANSFER, this call suspends the calling device process, assigns it to p1, resumes (transfers to) the suspended process p2, and in addition causes the interrupt transfer occurring upon device completion to assign the interrupted process to p2 and to resume the device process p1. va is the interrupt vector address assigned to the device. The procedure IOTRANSFER must be imported, and should be considered as PDP-11 implementation-specific.

It is necessary that interrupts can be postponed (disabled) at certain times, e.g. when variables common to the cooperating processes are accessed, or when other, possibly time-critical operations have priority. Therefore, every module is given a certain priority level, and every device capable of interrupting is given a priority level. Execution of a program can be interrupted, if and only if the interrupting device has a priority that is greater than the priority level of the module containing the statement currently being executed. Whereas the device priority is defined by the hardware, the priority level of each module is specified by its heading. If an explicit specification is absent, the level in any procedure is that of the calling program. IOTRANSFER must be used within modules with a specified priority only.

14. Compilation units

A text which is accepted by the compiler as a unit is called a *compilation unit*. There are three kinds of compilation units: main modules, definition modules, and implementation modules. A main module constitutes a main program and consists of a so-called *program module*. In particular, it has no export list. Imported objects are defined in other (separately compiled) program parts which themselves are subdivided into two units, called definition module and implementation module.

The *definition module* specifies the names and properties of objects that are relevant to clients, i.e. other modules which import from it. The *implementation module* contains local objects and statements that need not be known to a client. In particular the definition module contains constant, type, and variable declarations, and specifications of procedure headings. The corresponding implementation module contains the complete procedure declarations, and possibly further declarations of objects not exported. Definition and implementation modules exist in pairs. Both may contain import lists, and all objects declared in the definition module are available in the corresponding implementation module without explicit import.

```
$ DefinitionModule = DEFINITION MODULE Ident ":"
$ {import} {definition} END Ident ":" . ↑
$ definition = CONST {ConstantDeclaration ":"} |
$ TYPE {Ident ["=" type] ":"} |
```

```
$ VAR {VariableDeclaration ":"} |
$ ProcedureHeading ":".
$ ProgramModule = MODULE Ident [priority] ":" {import} block Ident ":".
$ CompilationUnit = DefinitionModule | [IMPLEMENTATION] ProgramModule.
```

The definition module evidently represents the interface between the implementation module on one side and its clients on the other side. The definition module contains those declarations which are relevant to the client modules, and presumably no other ones. Hence, the definition module acts as the implementation module's (extended) export list, and all its declared objects are exported.

Definition modules imply the use of qualified export. Type definitions may consist of the full specification of the type (in this case its export is said to be transparent), or they may consist of the type identifier only. In this case the full specification must appear in the corresponding implementation module, and its export is said to be *opaque*. The type is known in the importing client modules by its name only, and all its properties are hidden. Therefore, procedures operating on operands of this type, and in particular operating on its components, must be defined in the same implementation module which hides the type's properties. Opaque export is restricted to pointers. Assignment and test for equality are applicable to all opaque types.

As in local modules, the body of an implementation module acts as an initialization facility for its local objects. Before its execution, the imported modules are initialized in the order in which they are listed. If circular references occur among modules, their order of initialization is not defined.

SET	30				
THEN	57	56			
TO	64	31			
TYPE	87	72			
UNTIL	63				
VAR	88	77	73	34	33
WHILE	62				
WITH	67				
[81	36	20		
{	46				
	59	25			
]	81	36	20		
}	46				
†	36				

† denotes revisions

Appendix 2

Standard Utility Modules

The subsequently listed modules have proven to be of general usefulness for a wide range of applications. In particular, they concern the subject of input and output. The module *Terminal* represents a standard alphanumeric terminal for input and output. *FileSystem* represents the necessary operations to generate, read, write, name, and delete files organized as streams of characters or words.

The modules *Windows*, *TextWindows*, and *GraphicWindows* form a hierarchy of utilities for use on a high-resolution display. The latter two both rely on the basic module *Windows*. Closely connected with these window handlers are the modules *CursorMouse* and *Menu*. The former assumes the presence of a pointing device for input of coordinate values, a so-called mouse, and it causes its current position to be reflected on the display by a cursor. The module *Menu* relates the mouse with the display by providing a general command input facility in the form of so-called pop-up menus.

These modules are presented here in the form of definition modules. We emphasize that they are *not* part of the definition of the *language* Modula-2, and it is recognized that different implementations may either vary in details or in the selection of modules provided.

```

DEFINITION MODULE Terminal; (*S.E. Knudsen*)
PROCEDURE Read(VAR ch: CHAR);
PROCEDURE BusyRead(VAR ch: CHAR); (*returns 0C, if no character was typed*)
PROCEDURE ReadAgain; (*causes the last character read to be returned again
upon the next call of Read*)
PROCEDURE Write(ch: CHAR);
PROCEDURE WriteLn; (*terminate line*)
PROCEDURE WriteString(s: ARRAY OF CHAR);
END Terminal.

```


DEFINITION MODULE FileSystem; (*S.E. Knudsen*)
FROM SYSTEM IMPORT ADDRESS, WORD;

TYPE

Response = (done, notdone, notsupported, callerror,
unknownmedium, unknownfile, paramerror,
toomanyfiles, eom, deviceoff,
softparityerror, softprotected, softerror,
hardparityerror, hardprotected, timeout, harderror);

Command = (create, open, close, lookup, rename,
setread, setwrite, setmodify, setopen,
dolo, setpos, getpos, length,
setprotect, getprotect, setpermanent, getpermanent,
getinternal);

Flag = (er, ef, rd, wr, ag, bytemode);

FlagSet = SET OF Flag;

File = RECORD res: Response;
bufa, ela, ina, topa: ADDRESS;
elodd, inodd, eof: BOOLEAN;
flags: FlagSet;
CASE com: Command OF
create, open, getinternal: fileno, versionno: CARDINAL;
lookup: new: BOOLEAN;
setpos, getpos, length: highpos, lowpos: CARDINAL;
setprotect, getprotect: wrprotect: BOOLEAN;
setpermanent, getpermanent: on: BOOLEAN
END;
END;

- (* The routines defined by the file system can be grouped in routines for
1. Opening, closing and renaming of files.
(Create, Close, Lookup, Rename)
 2. Reading and writing of files.
(SetRead, SetWrite, SetModify, SetOpen, Dolo)
 3. Positioning of files.
(SetPos, GetPos, Length)
 4. Streamlike handling of files.
(Reset, Again, ReadWord, WriteWord, ReadChar, WriteChar) *)

PROCEDURE Create(VAR f: File; mediumname: ARRAY OF CHAR);
(* creates a new temporary (or nameless) file f on the named device. *)

PROCEDURE Close(VAR f: File);
(* terminates the operations on file f, i.e. cuts off the
connection between variable f and the file system. A temporary
file will hereby be destroyed whereas a file with
a not empty name remains in the directory for later use. *)

PROCEDURE Lookup(VAR f: File; filename: ARRAY OF CHAR; new: BOOLEAN);
(* searches file 'filename'. If the file does not exist and 'new' is TRUE,
a new file with the given name will be created. *)

PROCEDURE Rename(VAR f: File; filename: ARRAY OF CHAR);
(* changes the name of the file to 'filename'. If the new
name is empty, f is changed to be a temporary file. *)

PROCEDURE SetRead(VAR f: File);
(* initializes the file for reading. *)

PROCEDURE SetWrite(VAR f: File);
(* initializes the file for writing. *)

PROCEDURE SetModify(VAR f: File);
(* initializes the file for modifying. *)

PROCEDURE SetOpen(VAR f: File);
(* terminates any input- or output operations on the file. *)

PROCEDURE Dolo(VAR f: File);
(* is used in connection with SetRead, SetWrite and
SetModify in order to read, write or modify a file
sequentially. *)

PROCEDURE SetPos(VAR f: File; highpos, lowpos: CARDINAL);
(* sets the current position of file f to byte
 $\text{highpos} \cdot 2^{16} + \text{lowpos}$. *)

PROCEDURE GetPos(VAR f: File; VAR highpos, lowpos: CARDINAL);
(* returns the current byte position of file f. *)

PROCEDURE Length(VAR f: File; VAR highpos, lowpos: CARDINAL);
(* returns the length of file f in highpos and lowpos. *)

PROCEDURE Reset(VAR f: File);
(* sets the file into state opened and the position
to the beginning of the file. *)

PROCEDURE Again(VAR f: File);
(* prevents a subsequent call of ReadWord (or ReadChar)
from reading the next value on the file. Instead, the
value read just before the call of Again is returned
once more. *)

PROCEDURE ReadWord(VAR f: File; VAR w: WORD);
(* reads the next word on the file. *)

PROCEDURE WriteWord(VAR f: File; w: WORD);
(* appends word w to the file. *)

PROCEDURE ReadChar(VAR f: File; VAR ch: CHAR);
(* reads the next character on the file. *)

PROCEDURE WriteChar(VAR f: File; ch: CHAR);
(* appends character ch to the file. *)

END FileSystem.

```

DEFINITION MODULE InOut; (*N. Wirth*)
CONST EOL = 36C;
VAR Done: BOOLEAN;
    termCH: CHAR;

PROCEDURE OpenInput(defext: ARRAY OF CHAR);
(*request a file name and open input file "in".
  Done := "file was successfully opened".
  If open, subsequent input is read from this file.
  If name ends with ".", append extension defext*)

PROCEDURE OpenOutput(defext: ARRAY OF CHAR);
(*request a file name and open output file "out"
  Done := "file was successfully opened".
  If open, subsequent output is written on this file*)

PROCEDURE CloseInput;
(*closes input file; returns input to terminal*)

PROCEDURE CloseOutput;
(*closes output file; returns output to terminal*)

PROCEDURE Read(VAR ch: CHAR);
(*Done := NOT in.eof*)

PROCEDURE ReadString(VAR s: ARRAY OF CHAR);
(*read string, i.e. sequence of characters not containing
  blanks nor control characters; leading blanks are ignored.
  Input is terminated by any character < " ";
  this character is assigned to termCH.
  DEL is used for backspacing when input from terminal*)

PROCEDURE ReadInt(VAR x: INTEGER);
(*read string and convert to integer. Syntax:
  integer = ["+"|" "-] digit {digit}.
  Leading blanks are ignored.
  Done := "integer was read"*)

PROCEDURE ReadCard(VAR x: CARDINAL);
(*read string and convert to cardinal. Syntax:
  cardinal = digit {digit}.
  Leading blanks are ignored.
  Done := "cardinal was read"*)

PROCEDURE Write(ch: CHAR);

PROCEDURE WriteLn; (*terminate line*)

PROCEDURE WriteString(s: ARRAY OF CHAR);

PROCEDURE WriteInt(x: INTEGER; n: CARDINAL);
(*write integer x with (at least) n characters on file "out".
  If n is greater than the number of digits needed,
  blanks are added preceding the number*)

```

```

PROCEDURE WriteCard(x: CARDINAL);
PROCEDURE WriteOct(x: CARDINAL);
PROCEDURE WriteHex(x: CARDINAL);
END InOut.

```

```

DEFINITION MODULE RealInOut; (*N. Wirth*)
VAR Done: BOOLEAN;

PROCEDURE ReadReal(VAR x: REAL);
(*Read REAL number x according to syntax:
  ["+"|" "-] digit {digit} ["." digit {digit}]
  ["E"|"+"|" "-] digit {digit}

  Done := "a number was read".
  At most 7 digits are significant, leading zeroes not
  counting. Maximum exponent is 38. Input terminates
  with a blank or any control character. DEL is used
  for backspacing*)

PROCEDURE WriteReal(x: REAL; n: CARDINAL);
(*Write x using n characters. If fewer than n characters
  are needed, leading blanks are inserted*)

PROCEDURE WriteRealOct(x: REAL);
(*Write x in octal form with exponent and mantissa*)

END RealInOut.

```

the digits of "cmd" (from right to left),
interpreted as an octal number*)

END Menu.

DEFINITION MODULE Storage; (*SEK 5.10.80*)

FROM SYSTEM IMPORT ADDRESS;

PROCEDURE ALLOCATE(VAR a: ADDRESS; size: CARDINAL);

(* ALLOCATE allocates an area of the given size and returns
its address in a. If no space is available, the calling
program is killed. *)

PROCEDURE DEALLOCATE(VAR a: ADDRESS; size: CARDINAL);

(* DEALLOCATE frees the area at address a with the given size. *)

PROCEDURE Available(size: CARDINAL): BOOLEAN;

(* Available returns TRUE if size words could be allocated. *)

END Storage.

DEFINITION MODULE MathLib0;

(*standard functions; J.Waldvogel/N.Wirth, 10.12.80*)

PROCEDURE sqrt(x: REAL): REAL;

PROCEDURE exp(x: REAL): REAL;

PROCEDURE ln(x: REAL): REAL;

PROCEDURE sin(x: REAL): REAL;

PROCEDURE cos(x: REAL): REAL;

PROCEDURE arctan(x: REAL): REAL;

PROCEDURE real(x: INTEGER): REAL;

PROCEDURE entier(x: REAL): INTEGER;

END MathLib0.

Appendix 3

The ASCII Character Set

	0	20	40	60	100	120	140	160
0	nul	die		0	@	P	'	p
1	soh	dcl	!	1	A	Q	a	q
2	stx	dc2	"	2	B	R	b	r
3	etx	dc3	#	3	C	S	c	s
4	eot	dc4	\$	4	D	T	d	t
5	enq	nak	%	5	E	U	e	u
6	ack	syn	&	6	F	V	f	v
7	bel	etb	'	7	G	W	g	w
10	bs	can	(8	H	X	h	x
11	ht	em)	9	I	Y	i	y
12	lf	sub	*	:	J	Z	j	z
13	vt	esc	+	;	K	[k	{
14	ff	fs	.	<	L	\	l	
15	cr	gs	-	=	M]	m	}
16	so	rs	.	>	N	+	n	~
17	si	us	/	?	O	+	o	del

Layout characters

bs	backspace
ht	horizontal tabulator
lf	line feed
vt	vertical tabulator
ff	form feed
cr	carriage return

Separator characters

fs	file separator
gs	group separator
rs	record separator
us	unit separator

